MONASH University

Information Technology

# FIT1008/2085
# MIPS – Function Call

Prepared by:
Maria Garcia de la Banda
Revised by D. Albrecht, J. Garcia





# Where are we up to?

- **Discussed MIPS architecture and main design decisions**
- **Gone through the instruction set that we will use in this unit**
- **Practiced translation of decisions**
  - selection (if) and
  - loops (while and for)
- **Discussed how to create and access arrays of integers**
- **Discussed how variables are stored and accessed**
  - Global variables in the data segment
  - Local variables in the stack segment (the system stack)
    - Memory diagrams
    - Allocation/deallocation with `$sp`
    - Access by a negative offset from `$fp`

# Learning objectives for this lesson

- **To understand how functions are implemented in MIPS**
- **In particular:**
  - Use of the `jal` and `jr` instructions
  - Use of the system stack to satisfy function properties
- **To understand the reasons behind the decisions taken by the function calling convention**
- **To understand what a stack frame is, and its purpose**
- **To be able to implement a function call in MIPS**

MONASH University

# What is so special about functions?

# Function calling: return where?



# Function calling in MIPS

- **Remember: `jal` instruction for calling a function**
  - `jal` is like jump (`j`), but it first saves the address of the instruction following the `jal` ( ie. at address =PC+4 ) in register `$ra`



# Function calling in MIPS

- **Remember: `jr` instruction for returning from a function**
  - `jr` jumps to the address held in the given GPR register (usually `$ra`)



# Function calling in MIPS – first go

- **To write a function**
  - Put label at the start of the function
  - Write body of the function
  - End function with `jr $ra`
- **To call a function**
  - Write `jal label`
  - When the function returns, program will continue from the next instruction
- **Is this really all it is needed to call a function?**
  - What are we forgetting?

# Passing and returning data – second go

- **Some functions have parameters**
  - Need a way of passing values for these parameters (i.e., arguments) from caller to function
- **Some functions return values**
  - Need a way of getting the return value safely back to caller
- **Solution: reserve some registers for these tasks**
  - Can extend the "syscall" data passing method:
  - Pass function parameters in `$a0`, `$a1`, `$a2`, `$a3`
  - Return values in `$v0`, `$v1`
- **This convention still has some problems**
  - Will make a better system later
  - But let's see how it works for now

Focus on the blue parts

# Example

```
# Part of program that uses
# function fact

        …
        # Read int from user
        addi  $v0, $0, 5
        syscall

        # Pass parameter in $a0
        addi  $a0, $v0, 0
        # Call!
        jal   fact
        # Put result in $t1
        addi  $t1, $v0, 0

        # Print factorial
        addi  $v0, $0, 1
        addi  $a0, $t1, 0
        syscall

        # Exit
        addi  $v0, $0, 10
        syscall
```

```
# Factorial function

fact:   addi  $t0, $0, 1  # result=1

        # Repeat while parameter
        # in $a0 is > 1
loop:   addi $t1, $0, 1
        slt $t1, $t1, $a0 # is > 1?
        beq $t1, $0, end  # if not go end

        # Multiply result by
        # parameter
        mult  $t0, $a0
        mflo $t0

        # Decrement parameter
        addi  $a0, $a0, -1

        # loop back to test
        j  loop

        # At end, $t0 contains
        # factorial of user input
end:    addi $v0, $t0, 0 #$v0=answer
        jr  $ra          # Return
```

# Limitations of our second go

- **This simple function-calling convention works, but has limits**
- **Function must:**
  - Not call other functions (i.e., it is a “leaf function”)
  - Not use more than four parameters ($a0-$a3)
  - Only write to “safe” registers $v0-$v1, $a0-$a3, $t0-$t9
  - Not have/use local variables (did not consider $sp and $fp)
- **For more sophisticated function calling, one needs a more sophisticated convention**

MONASH University

# Limitations of our naïve approach

# Limitation: calling other functions

- **Return address register ($ra) can hold only one value**



# Limitation: calling other functions

- **Return address register (`$ra`) can hold only one value**
- **Solution: save and restore `$ra` register on the stack upon function entry/exit (push on entry, pop on exit)**



# Limitation: passing arguments

- **Not enough registers (4) to pass arguments to some functions**

```
addi $a0, $0, 1
lw $a1, x
addi $a2, $0, 0
lw $a3, -4($fp)
addi $??, $0, 2
jal five
...
```

no such register $a4

```
five: # takes 5
      # parameters
      ...
      # examine
      # $a0, etc
      ...
      jr $ra
```

# Limitation: passing arguments

- **Not enough registers (4) to pass arguments to some functions**
- **Solution: pass arguments on stack**

```
# push 2
# push global y
# push 0
# push local x
# push 1
jal five
# pop
# pop
# pop
# pop
# pop
```




```
five: # takes 5
      # parameters
      ...
      # examine
      # stack
      ...
      jr $ra
```

For simplicity, in FIT1008/2085 we will pass all arguments onto the stack

# Limitation: saving registers

- **Function may use registers which hold important values**

```
...
lw $t0, a
...
...
jal func
...
...
# $t0 has been
# changed!
add $t0, $t0, $v0
...
```

```
func: ...
      # trashes
      # $t0
      lw $t0, x
      ...
      jr $ra
```

- **Solution: save/restore registers on stack (not in FIT1008/FIT2085)**
- **While you will not be doing this; assemblers do!**

# Limitation: Local variables (reminder)

- **A function needs its own local variable storage**

```
def main():

  m1 = 5
  m2 = -42
   ...
  func()
```




```
def func():

  f1 = 87
  ...
```

# Limitation: Local variables (reminder)

- A function needs its own local variable storage
- Solution: function adjusts `$fp` to top of stack, then allocates its own locals



# Limitation: Local variables – restore $fp

- **On function return, stack state (including `$fp`) must be restored**

```
def main():

  m1 = 5
  m2 = -42
   ...
  func()
```




```
def func():

  f1 = 87
  ...
```

# Limitation: Local variables – restore $fp

- On function return, stack state (including `$fp`) must be restored
- Solution: save/restore `$fp` on stack

```
def main():

  m1 = 5
  m2 = -42
   ...
  func()
```




```
def func():
  # save $fp
  f1 = 87
  ...
  # restore $fp
```

# Function calling convention

| Task | Whose responsibility? |
|---|---|
| save / restore temporary registers | caller |
| pass / clear arguments | caller |
| save / restore `$ra` | callee |
| save / restore `$fp` | callee |
| allocate / deallocate local variables | callee |

Convention; both can clear

Convention; both can do this

# Function calling convention

**When calling a function, caller:**

1. Saves temporary registers by pushing their values on stack
2. Pushes arguments on stack
3. Calls the function with `jal` instruction

**On function entry, callee:**

1. Saves `$ra` by pushing its value on stack
2. Saves `$fp` by pushing its value on stack
3. Copies `$sp` to `$fp`
4. Allocates local variables


# Example: Function Call (caller point of view)

# Example: caller

```
""" Illustrating function call. """

def power(base:int, exp:int) -> int:
    ...
    return result

def main():

    base, exp, result = 0, 0 ,0
    read(base)
    read(exp)

    result = power(base, exp)

    print(result)
```

base is at -12($fp)

exp is at -8($fp)

result is at -4($fp)




This is the memory diagram at this point

Assume user has entered 3 for base and 4 for exp

# Example: caller

```
""" Illustrating function call. """

def power(base:int, exp:int) -> int:
    ...
    return result

def main():
    base, exp, result = 0, 0 ,0
    read(base)
    read(exp)

    result = power(base, exp)

    print(result)
```

base is at -12($fp)
exp is at -8($fp)
result is at -4($fp)

Wait, we forgot to save $ra, $fp, no?

No: `main` is not meant to return anywhere, just to exit, so no need

```
        .text
main:   # set $fp and make space for locals
        addi $fp, $sp, 0    # copy $sp into $fp
        addi $sp, $sp, -12 # 3 locals = 12bytes

        # Initialize locals
        sw $0, -12($fp) # base=0
        sw $0, -8($fp)  # exp=0
        sw $0, -4($fp)  # result=0

        # read(base)
        addi $v0, $0, 5
        syscall
        sw $v0, -12($fp)# base=$v0

        # read(exp)
        addi $v0, $0, 5
        syscall
        sw $v0, -8($fp) # exp=$v0

# Now we are up to the
# function call ...
```

# Example: caller

caller step 1: save temporary registers by pushing their values on stack

not needed in FIT1008/FIT2085



# Example: caller

caller step 2: push function arguments onto the stack

two arguments, called base and exp

```
def power(base, exp):
```



# Example: caller

note offsets of arguments:

base at 0($sp)

exp at 4($sp)




caller step 2: push function arguments onto the stack

two arguments, called base and exp

```
def power(base, exp):
```

# Example: caller

caller step 3: call function with `jal` instruction

no effect visible on the stack



# Example: caller

```
""" Illustrating function call. """

def power(base:int, exp:int) -> int:
    ...
    return result

def main():

    base, exp, result = 0, 0 ,0
    read(base)
    read(exp)

    result = power(base, exp)

    print(result)
```

base at -12($fp)
exp at -8($fp)
result at -4($fp)
arg1 at 0($sp)
arg2 at 4($sp)

```
# ... continued from
# previous slide

    # Call power(base,exp)
    # push 2 * 4 = 8 bytes
    # of arguments
    addi $sp, $sp, -8

    # arg 1 = base
    lw $t0, -12($fp) # load base
    sw $t0, 0($sp)   # arg 1=base

    # arg 2 = exp
    lw $t0, -8($fp)  # load exp
    sw $t0, 4($sp)   # arg 2=exp

    # link and goto power
    jal power


# main function To Be
# Continued next lecture ...
```


# Example: Function Entry (callee point of view)

# Example: callee

```
""" Illustrating also function return. """

def power(base:int, exp:int) -> int:
    """ Computes base to the power of exp. """
    result = 1
    while exp > 0:
        result *= base
        exp -= 1
    return result
```

# Example: callee




callee steps 1 and 2: save $ra and $fp by pushing their values on stack

can do both these steps at once

# Example: callee

saved $fp contains address of another location on stack

callee steps 1 and 2: save $ra and $fp by pushing their values on stack

can do both these steps at once



# Example: callee




callee step 3: copy $sp into $fp

now main's local variables are inaccessible

# Example: callee




callee step 4: allocate local variables

in this function, one local variable (`result`)

# Example: callee




callee step 4: allocate local variables

in this function, one local variable (result)

# Frame Pointers



# Stack Frames

the data on the stack that is associated with a function is called a stack frame



# Stack Frames



# Stack Frames



# Local variables

result is at -4($fp)

power's local variables are accessed as main's were, with negative offsets from $fp



# Function parameters



# Example: callee

```
result is at -4($fp)
base is at 8($fp)
exp is at 12($fp)
```

```
""" Illustrating also function return. """

def power(base:int, exp:int) -> int:
    """ Computes base to the power of exp. """
    result = 1
    while exp > 0:
        result *= base
        exp -= 1
    return result
```

```
power:  # Save $ra and $fp in stack
        addi $sp, $sp, -8 # make space
        sw $ra, 4($sp)    # save $ra
        sw $fp, 0($sp)    # save $fp

        # Copy $sp to $fp
        addi $fp, $sp, 0

        # Alloc local variables
        # 1 * 4 = 4 bytes.
        addi $sp, $sp, -4

        # Initialize locals.
        addi $t0, $0, 1
        sw $t0, -4($fp)  # result=1

        # Now we are inside
        # the function body.
```

# Example: callee

```
result is at -4($fp)
base is at 8($fp)
exp is at 12($fp)
```

```
""" Illustrating also function return. """

def power(base:int, exp:int) -> int:
    """ Computes base to the power of exp. """
    result = 1
    while exp > 0:
        result *= base
        exp -= 1
    return result
```

```
loop:   # Stop if not (exp > 0)
        lw $t0, 12($fp)  # load exp
        slt $t0, $0, $t0 # is 0 < exp?
        beq $t0, 0, end  # if not go to end

        # result *= base
        lw $t0, -4($fp) # load result
        lw $t1, 8($fp)  # load base
        mult $t0, $t1   # base*result
        mflo $t0
        sw $t0, -4($fp) # result=base*result

        # exp -= 1
        lw $t0, 12($fp)  # load exp
        addi $t0, $t0, -1
        sw $t0, 12($fp)  # exp = exp-1

        # Repeat loop.
        j loop

end:    # Now ready to return.
# Continued in next lecture ...
```

# Summary

- **Function calling**
  - `jal` and `jr` instructions
- **Function calling convention**
  - For making a function call
- **Structure of the stack**
  - stack frames